

```
> n:=3;
D(y) := t -> f(y(t));
alias(F=f(y(0)),F[y]=D(f)(y(0)),F[yy]=(D@@2)(f)(y(0)));
```

$$n := 3$$

$$D(y) := t \rightarrow f(y(t))$$

$$F, F_y, F_{yy}$$

```
> ye:=taylor(y(dt),dt,n+1);
```

$$ye := y(0) + F dt + \frac{1}{2} F_y F dt^2 + \left(\frac{1}{6} F_{yy} F^2 + \frac{1}{6} F_y^2 F \right) dt^3 + O(dt^4)$$

```
> Y[1]:=y(0);
Y[2]:=taylor(y(0)+dt*a[2,1]*f(Y[1]),dt,n+1);
Y[3]:=taylor(y(0)+dt*sum(a[3,j]*f(Y[j]),j=1..2),dt,n+1);
```

$$Y_1 := y(0)$$

$$Y_2 := y(0) + a_{2,1} F dt$$

$$Y_3 := y(0) + (a_{3,1} F + a_{3,2} F) dt + a_{3,2} F_y a_{2,1} F dt^2 + \frac{1}{2} a_{3,2} F_{yy} a_{2,1}^2 F^2 dt^3 + O(dt^4)$$

```
> y1:=taylor(y(0)+dt*sum(b[j]*f(Y[j]),j=1..n),dt,n+1);
```

$$y1 := y(0) + (b_1 F + b_2 F + b_3 F) dt + (b_3 F_y F (a_{3,1} + a_{3,2}) + b_2 F_y a_{2,1} F) dt^2 +$$

$$\left(b_3 \left(a_{3,2} F_y^2 a_{2,1} F + \frac{1}{2} F_{yy} a_{3,1}^2 F^2 + F_{yy} a_{3,1} F^2 a_{3,2} + \frac{1}{2} F_{yy} a_{3,2}^2 F^2 \right) + \frac{1}{2} b_2 F_{yy} a_{2,1}^2 F^2 \right) dt^3 +$$

$$O(dt^4)$$

```
> tau:=convert(taylor(ye-y1,dt),polynom);
```

$$\tau := (-b_1 F - b_2 F - b_3 F + F) dt + \left(-b_3 F_y F (a_{3,1} + a_{3,2}) - b_2 F_y a_{2,1} F + \frac{1}{2} F_y F \right) dt^2 + \left(-b_3 \left(a_{3,2} F_y^2 a_{2,1} F + \frac{1}{2} F_{yy} a_{3,1}^2 F^2 + F_{yy} a_{3,1} F^2 a_{3,2} + \frac{1}{2} F_{yy} a_{3,2}^2 F^2 \right) - \frac{1}{2} b_2 F_{yy} a_{2,1}^2 F^2 + \frac{1}{6} F_{yy} F^2 + \frac{1}{6} F_y^2 F \right) dt^3$$

```
> eqns:={coeffs(expand(tau),[dt,F,F[y],F[yy]])};
```

$$eqns := \{-b_3 a_{3,1} - b_3 a_{3,2} - b_2 a_{2,1} + \frac{1}{2}, -b_3 + 1 - b_1 - b_2, \frac{1}{6} - b_3 a_{3,2} a_{2,1},$$

$$\frac{1}{6} - \frac{1}{2} b_3 a_{3,1}^2 - b_3 a_{3,1} a_{3,2} - \frac{1}{2} b_3 a_{3,2}^2 - \frac{1}{2} b_2 a_{2,1}^2 \}$$

```
> vars:=indets(eqns);
```

```

vars := {a2,1, b1, b2, b3, a3,1, a3,2}
> solve(eqns, vars);
{a3,1 =  $\frac{\text{RootOf}(-18 b_3 a_{2,1}^2 + 18 b_3 a_{2,1}^3 + 1 + (-36 b_3 a_{2,1}^2 + 12) _Z + 36 _Z^2, \text{label} = \_L2)}{a_{2,1} b_3}$ , b2 =
 $\frac{1}{6}(-6 \text{RootOf}(-18 b_3 a_{2,1}^2 + 18 b_3 a_{2,1}^3 + 1 + (-36 b_3 a_{2,1}^2 + 12) _Z + 36 _Z^2, \text{label} = \_L2) - 1$ 
+ 3 a2,1) / a2,12, a2,1 = a2,1, b1 = - $\frac{1}{6}(6 b_3 a_{2,1}^2 - 6 a_{2,1}^2$ 
- 6 RootOf(-18 b3 a2,12 + 18 b3 a2,13 + 1 + (-36 b3 a2,12 + 12) _Z + 36 _Z2, label = _L2) - 1
+ 3 a2,1) / a2,12, a3,2 =  $\frac{1}{6} \frac{1}{b_3 a_{2,1}}$ , b3 = b3}
> assign(%);
> b[3] := 3/4;
a[2,1] := 1/3;

b3 :=  $\frac{3}{4}$ 
a2,1 :=  $\frac{1}{3}$ 
> allvalues(b[1]); allvalues(b[2]); allvalues(a[3,1]); a[3,2];

 $\frac{1}{4}, -2$ 
0,  $\frac{9}{4}$ 
0, -1
 $\frac{2}{3}$ 
>

```